SP5.28: Advanced Visual Programming

OLE Automation
SP5.28 Advanced Visual Programming

OLE Automation

Object Linking and Embedding enables your applications to use external data by either linking to it or by actually embedding it into your program.  The distinction between the two is important.

Linked Objects 

A linked object in your application only holds a pointer to the original data.  An embedded object, however, can be activated within your program using a double-click.  Your application’s menu is then replaced by the embedded application’s menu (e.g. Excel) and to all intents and purposes, it is just as if you had transferred to that application.  For a linked object, any changes made are to the original object;  for embedded objects, VB loads a snap-shot of the object at run-time, and any changes are only made to the copy.

OLE Automation

Many OLE 2 applications support an OLE feature called OLE Automation.  Such programs export methods and properties, and possibly functions, which you can use in your code.  For example, a spreadsheet application may expose a worksheet, chart, cell, or range of cells all as different types of objects.  A word processor might expose objects such as application, document, paragraph, sentence, bookmark, or selection.

When an application supports OLE Automation, the objects it exposes can be accessed by Visual Basic.  For example, you could write code to create and/or modify Excel data, and use Excel’s built-in functions to work on those data without Excel itself being visible.  This is where Microsoft’s vision of VB as the ‘glue’ between components begins to be realised.  You can build applications from the exposed objects of other applications, using features and function built into those programs to operate on your data for you.
Using OLE Automation Objects

Like any other object you use in VB, OLE automation objects come complete with properties and methods that you can use in your VB code.  If you want to make use of the statistical functions in Excel, for example, then simply open an Excel sheet object and make use of its properties.  OLE Automation is available between any two compliant applications, providing at least one of them has a scripting language able to make use of the facility!  You can OLE Automate WORD, for example, from either EXCEL (using VBA) or from within a VB stand-alone program.

There isn’t all that much to learn about the VB side of OLE automation.  Since you are dealing with objects provided by applications outside of Visual Basic, the real trick in mastering it is knowing what facilities these objects provide you with.

Using OLE Automation is a three-stage process:

1. Create a new instance of the server application using the New keyword (see later)

2. Use the server application’s objects by invoking methods and properties, just as in Excel, say.

3. Quit the server object (using the Quit method) and release the memory used by setting the application reference to Nothing when you are done.

Seeing what’s available

VB has a built-in object browser facility (press F2, or View, Object Browser from the main menu).  This shows you all the objects currently in your VB application, and those to which your code has access, even if it is not currently making use of them.

You can extend this list of available objects by selecting other libraries to appear in the Browser window by setting up a reference to that object from the References item in the Project menu.    From the menu, select “Project” and “References”.  Look for the “Microsoft Excel 8.0 Object Library”.  Version 8.0 relates to Office97; for Office2000 use the Excel 9.0 library.  (The Excel 8.0 object library has been selected in the illustration below).  

Creating References to OLE Applications

Once you have added the Excel 8.0 Object Library as described above, you can use the object browser to view the methods and properties within:

Steps to OLE Automation

When you refer to another application by using OLE Automation, you actually refer to one of the application’s top-level objects.  The task you want to accomplish determines the object you refer to.  The following table shows some common top-level objects:

	To return
	From this application
	Use this class name

	A reference to the MS Project Application object
	Microsoft Project
	MSProject.Application

	A reference to the MS Project Project object
	Microsoft Project
	MSProject.Project

	A reference to the MS Word Application object - allows you to run Word VBA statements
	MS Word
	Word.Application

	A reference to the MS Excel Application object
	MS Excel
	Excel.Application

	A reference to the MS Excel Sheet object
	MS Excel
	Excel.Sheet

	A reference to the MS Excel Chart object
	MS Excel
	Excel.Chart


The NEW method starts a new instance of an application invisibly and returns a reference to a new top-level object.  For example, the following code loads MS Project and returns the MS Project Application object:

Dim projapp as Object

Set projapp=New MSProject.Application

The New function returns a reference to a top-level object from an application that’s already running. Having set up a reference to the Excel object, in one of the ways described above, you have established a connection between your user application (the OLE client) and Excel (the OLE server).

You now need to know some things about the application you intend to control (we will use Excel as an example), such as what objects you want to control and how to refer to them.  This requires a knowledge of the object hierarchy.

The Object Hierarchy

Each object that supports OLE automation is part of a logical ‘tree’ of objects within the host application.  For an object at a certain level within the hierarchy to exist, all higher levels must also exist.  For example, you cannot have an Excel worksheet object without having an Excel Workbook and the Excel application itself.

Exercise 1: Creating OLE Objects
Create a new project, and add the declaration for the new object to the General Declarations section:


Dim xlApp as Excel.Application


Dim xlBook as Excel.Workbook


Dim xlSheet as Excel.Worksheet
Add the following to the Form_Load procedure:

Set xlApp = New Excel.Application


Set xlBook = xlApp.Workbooks.Add


Set xlSheet = xlBook.Activesheet


xlApp.Visible = True
Try running the program - you will see that Excel is loaded and displayed by your VB program.  If you close down the Excel worksheet, you will see that your VB program is still running.

The New method generates a new instance of the root OLE automation object (Excel) as soon as a dependent object is requested (in this case the workbook).  It ‘knows’ what the root object is - you don’t need to worry about that!  It follows logically that there must also be some method internal to the Excel.Application object that enables us to create a new worksheet without generating a new instance of Excel!  We will come to that later.
Controlling the Server Application (OLE1.VBP)

Create a new project in Visual Basic, and add a text box as shown in the illustration below.  You can if you wish also add a label as shown:



Add this code to the Declarations section:

Dim xlApp As Excel.Application


Dim xlBook As Excel.Workbook


Dim xlSheet As Excel.Worksheet
this to the Form_Load event:

Set xlApp = New Excel.Application

    
Set xlBook = xlApp.Workbooks.Add

    
Set xlSheet = xlBook.Worksheets.Add


xlApp.Visible = True

this to the Text1_Change event:

xlSheet.Cells(1, 1).Value = Text1.Text

and this to the Form_Unload event:

xlBook.Close savechanges:=False

    
xlApp.Quit

    
Set xlSheet = Nothing

    
Set xlBook = Nothing

    
Set xlApp = Nothing

This is what it should look like …

Adding New Sheets

As stated earlier, the Excel.Application object has an internal function that creates a new sheet - this is the Add method.  Each collection object has an Add method that creates a new member of the array.  To create a new worksheet, therefore, you would simply use the code:

xlApp.Worksheets.Add
where xlApp has been defined as an Excel object.
Working with Existing Objects

As its name suggests, New creates a new object in you code, but when you need to access a file that already exists, you should use the related command GetObject.
For example, suppose you wanted to open an Excel workbook called TEST.XLS.  The command for this would be:

  Set zFile = GetObject(“Test.XLS”, “Excel.Sheet”)

This would create an OLE automation link between your VB program and a specific file on disk, using Excel to maintain it.  No new object needs to be generated in this case.

Closing an Object

You may have noticed that after executing the previous code examples there are several copies of Excel running.  This is clearly undesirable and you need to explicitly close the server application when you have finished with it:

xlApp.Quit
Note that if you merely use 

Set xlSheet = Nothing
the sheet will disappear, but the application remains active.  This will not close down the whole application.

If you set an object variable to nothing, you are destroying that object, telling the OLE server application that you no longer need it to maintain the file that was created or opened.  The memory used by the object is released and the server application destroys the document you were using.  If appropriate, you should therefore remember to save the file before setting the object to nothing.

Exercise 2: Using Excel functions (OLE2.VBP)
This is a fairly simple program that lets you enter some numbers and uses Excel functions to carry out one of a range of arithmetic or statistical operations on them.  This illustrates the essential technique for making use of server applications via OLE automation.

Step 1
Open a new VB project and add controls to the form as illustrated below (or open the OLE2.VBP project provided!)

Step 2

Note there is a blank label near the bottom of the form.

Add the code as follows:

General declarations:

Dim xlApp As Excel.Application

Dim xlBook As Excel.Workbook

Dim xlSheet As Excel.Worksheet

Dim optsel As Integer

Private Sub cmdCalculate_Click()
  Dim nElement As Integer

  Label2.Visible = False

  MousePointer = 11

  For nElement = 0 To 5

    xlSheet.Cells(1 + nElement, 1).Value = Val(Text1(nElement))

  Next

  Select Case optsel

    Case 0   'add

        xlSheet.Cells(8, 1).Formula = "=Sum(A1:A6)"

    Case 1   'multiply

        xlSheet.Cells(8, 1).Formula = "=Product(A1:A6)"

    Case 2   'average

        xlSheet.Cells(8, 1).Formula = "=Average(A1:A6)"

    Case 3   'standard deviation

        xlSheet.Cells(8, 1).Formula = "=StDevP(A1:A6)"

  End Select

  MousePointer = 0

  Label2.Caption = "The result is : " & xlSheet.Cells(8, 1).Value

  Label2.Visible = True

End Sub

Private Sub cmdDisplayExcel_Click()
cmdDisplayExcel.Caption = "&Display Excel"

End Sub

Private Sub cmdExit_Click()
  End

End Sub

Private Sub Form_Load()
  Left = (Screen.Width - Width) / 2

  Top = (Screen.Height - Height) / 2

  Show

  DoEvents

  On Error GoTo OLE2ErrorHandler

  MousePointer = 11

  Label2.Caption = "Loading Excel ..."

  Set xlApp = New Excel.Application

  Set xlBook = xlApp.Workbooks.Add

  Set xlSheet = xlBook.Worksheets.Add

  MousePointer = 0

  On Error GoTo 0

  Label2.Caption = ""

  Exit Sub

OLE2ErrorHandler:

  On Error GoTo 0

  MsgBox "Could not create the Excel worksheet", vbExclamation, "error"

  End

End Sub
Private Sub Form_Unload(Cancel As Integer)
  xlBook.Close savechanges:=False

  xlApp.Quit

  Set xlApp = Nothing

  Set xlBook = Nothing

  Set xlSheet = Nothing

End Sub

Private Sub Option1_Click(Index As Integer)
  optsel = Index

End Sub

Try running the program.  You could try modifying it so that the “Display Excel” button changes to “Hide Excel” once the sheet is displayed, and a check is made to ensure that none of the 6 data entry cells are left blank when the “calculate” button is clicked.

Exercise 3: Interacting with Data in Excel (OLE3.VBP)
In this exercise, we will import data from an Excel worksheet, and carry out some simple analysis of the data.

Suppose that payments have been made to a number of individuals over a period of time, and a means of determining the total sum paid to each is required.  A data file (OLE3.XLS) has been set up containing some sample data:

We need to create a user interface in VB that looks something like this:

Add the following code:

Dim xlApp As Excel.Application

Dim xlBook As Excel.Workbook

Dim xlSheet As Excel.Worksheet

Dim startrow As Integer

Dim endrow As Integer

Dim namecol As Integer

Dim moneycol As Integer

Dim datecol As Integer

Private Sub butQuit_Click()

  End

End Sub

Private Sub Command1_Click()

  xlApp.Visible = True

  'zSheet.Parent.Windows(1).Visible = True

End Sub

Sub readdata()

  MousePointer = 11

  zfile = App.Path + "\ole3.xls"

  Label2.Caption = "Opening " & zfile

  Set xlApp = New Excel.Application

  Set xlBook = xlApp.Workbooks.Open(zfile)

  Set xlSheet = xlBook.Worksheets(1)

  Label2.Caption = ""

  MousePointer = 0

  startrow = 4

  namecol = 2

  datecol = 1

  moneycol = 3

  zrow = startrow

  Label2.Caption = "Reading names from data file"

  While xlSheet.Cells(zrow, namecol) <> ""

    If lisNames.ListCount = 0 Then

        lisNames.AddItem xlSheet.Cells(zrow, namecol)

        zrow = zrow + 1

    Else

        zname = xlSheet.Cells(zrow, namecol)

        gotit = False

        For k = 1 To lisNames.ListCount

            If zname = lisNames.List(k - 1) Then gotit = True

        Next k

        If Not gotit Then lisNames.AddItem zname

        zrow = zrow + 1

    End If

  Wend

  endrow = zrow - 1

  Label2.Caption = "There are " & lisNames.ListCount & " names"

End Sub

Private Sub Form_Load()

  Show

  DoEvents

  On Error GoTo errorhandler

  readdata

  Exit Sub

errorhandler:

  On Error GoTo 0

  MsgBox "error"

  End

End Sub

Private Sub Form_Unload(Cancel As Integer)

  xlApp.Quit

  xlBook.Close savechanges:=False

  xlApp.Quit

  Set xlApp = Nothing

  Set xlBook = Nothing

  Set xlSheet = Nothing

End Sub

Private Sub lisNames_Click()

  chosenname = lisNames.Text

  zsum = 0

  txt = ""

  For k = startrow To endrow

    zname = xlSheet.Cells(k, namecol)

    zcash = xlSheet.Cells(k, moneycol)

    zdate = xlSheet.Cells(k, datecol)

    If zname = chosenname Then

      zsum = zsum + zcash

      txt = txt+Format(zdate)+"  "+Format(zcash,"£###.00")+Chr$(13)+Chr$(10)

    End If

  Next k

  txt = txt + vbCrLf

  txt = txt + "Total = " + Format(zsum, "£###.00")

  txtSums.Text = txt

End Sub

The result would look something like this…

Exercise 4 : Interaction between other applications

In this exercise, we will look at automating Word.

Example 4.1 : An extremely simple illustration of manipulating the Word object from VB:

Add command buttons and a text box (set the multiline property to be true) to a new form and enter this code (the first line goes in the General Declarations section):

Dim wdDoc As Word.Document

Private Sub Command1_Click()

    Set wdDoc = New Word.Document

    wdDoc.Select

    Selection.InsertAfter Text:=Text1.Text

End Sub

Private Sub Command2_Click()

    End

End Sub

Private Sub Form_Unload(Cancel As Integer)

    wdDoc.Close savechanges:=False

    Set wdDoc = Nothing

End Sub

You can, of course change the text written to the document. 

Try: modifying the above to print out the text in a different font and in bold.

